
1 
 

Reading in Data Directly from Microsoft Word Questionnaire Forms 

ABSTRACT 

If someone comes to you with hundreds of questionnaire forms in Microsoft (MS) Word file format, and 

asks you to subtract the data from the forms into a SAS® dataset, you may have several ways to 

handle this. However, as a SAS programmer, the shortcut is to write a SAS program to read in the 

data directly from Word files into a SAS dataset. This paper will show how it can be done with simple 

SAS programming skills, such as using FILENAME with the PIPE option, DDE, function call 

EXECUTE( ),  and so on. 

 

INTRODUCTION 

MS Word file is one of the most popular things in the digital world. One study at our center (Center for 

Health Equity Research and Promotion, VA Pittsburgh Healthcare System) set the blank questionnaire 

template form in Word file and used it to collect data from each study subject. The task requested is to 

get the answers on questionnaire form into a SAS dataset, see Figure 1. 

 

 

 

 

 

 

 

 

 

 

 

Specifically, for the form above, the checkbox information needs to be collected as numeric values into 

a SAS dataset. However, the data in Word file is not organized in a table, and there are no 

recognizable bookmarks or labels in the file. First, searching for related publications online, I could not 

find any that fits this scenario. Then, I had a close look at the file and poked around, it turned out that I  

Figure 1: From raw data on MS Word questionnaire form to data in a SAS dataset 

got it done with several simple SAS programming skills. It takes three steps to move the raw 

information from MS Word files to a SAS dataset: 1. Copy all the Word files into a folder, 2. Convert 

Word files to RTF files, and 3) Subtract data from RTF files into a SAS dataset. The reason to convert 

Word files to RTF files is that SAS cannot read in and recognize the source code of Word files. If you 

feel that the code behind RTF file is difficult to read, then you will agree that the code behind Word file 

is impossible to make sense to a SAS programmer. Figure 2 shows the Notepad views of the 

screenshots of both Word and RTF files with only one sentence, “Hello, World!”. You cannot even find 

the sentence using Search function in Notepad view of the Word file source code. However, RTF 

source code can be read into SAS as character values, just like reading in text data file. Once the data 

file is read into SAS, it is like a package received, the next step is to unwrap the package and take out 

what we ordered. 

 

 

 

 

 

 

 

 

                                                                                 

 

 

Well, the text file is much simpler than RTF file and easier for SAS to process. Why not save the MS 

word file as a text file directly? Actually, this was what I tried first. But it does not work because all the 

formats and objects (including checkbox) are lost in the conversion, which will serve as indicators to 

identify the values to read. 

The coding environment for this paper: Win XP 2002 Service Pack 3, Microsoft Office Word 2007, and 

SAS 9.3.  

 

 

Figure 2: Notepad views of a MS Word and a RTF file 

Notepad view of a MS Word file Notepad view of a RTF file 

PROGRAMS  
 
1.  Get the list of all Word files in the folder 

 
***************** 

* Get file list * 

*****************; 

libname output "C:\Studies\IRB Survey\Word2SAS\SAS";    

 

%let wordPath=C:\Studies\IRB Survey\Word2SAS\Word\;    

     * &wordPath contains all Word questionnaire forms *; 

%let rtfPath=C:\Studies\IRB Survey\Word2SAS\rtf\; 

%let fileType=.docx;   * Or .doc *; 

 

filename files pipe "dir ""&wordPath*&fileType""";      ❶         

 

data fileList;                                                

   infile files lrecl=300 truncover;                           

   input line $200.;    

   retain fileID;                                           ❷ 

   if not index(line,"&fileType") then delete; 

 else fileID+1;  
   fileName = strip(substr(line,39,199)); 

   wordPathname="&wordPath"||left(fileName);    

   rtfPathname="&rtfPath"||left(tranwrd(fileName, "&fileType", ".rtf"));     

   keep fileID fileName wordPathname rtfPathname;                                                                                

run;   

 

 
Statements above FILENAME specify the location and input file extension information for rest of the 

program. The file extension before Word 2007 is ‘.doc’. 

The PIPE ❶ option in FILENAME statement, with file extension specified in command DIR, will collect 

file attribute information of all those files and sub-directory information in the folder specified, and save 
them in the default variable LINE. The contents of each line just looks like what we see in Window 
Explorer, including file name, size, type, etc. Then we can use function SUBSTR( ) to subtract the 

piece that we need. ❷ Four new variables (1. fileID – used to identify individual output SAS dataset, 2. 

filename – filename (including extension), 3. wordPathname – pathnames of input Word files, and 4. 
rtfPathname  - pathname of converted RFT files) are created from variable LINE in dataset fileList for 
the purpose of data manipulation later, see Figure 3.  
 
 

 

 

 
Figure 3: SAS dataset fileList 

2.  Convert Word files to RTF files 

 
***************************** 

* Convert Word to RTF files * 

*****************************; 

options noxwait noxsync; 

 

x call "C:\Program Files\Microsoft Office\Office12\winword.exe"; 

 

data _null_;                                          ❶ 

   wait=sleep(3); 

run; 

 

fileName wordLink dde 'WinWord|System'; 

 

%macro word2rtf(inPathname,outPathname); 

data _null_; 

   file wordLink; 

 

   put '[FileOpen.Name = "'"&inPathname"'"]'; 

   put '[FileSaveAs "'"&outPathname"'",6]';           ❷ 

   put '[FileClose]'; 

run; 

%mend word2rtf; 

 

data _null_; 

   set fileList; 

   call execute('%word2rtf('||wordPathname||', '||rtfPathname||')');   ❸ 

run; 

 

data _null_; 

   file wordLink; 

   put '[FileExit]';  

run; 

 

fileName wordLink clear; 

DDE (dynamic data exchange) connection between SAS and MS Word is established at the beginning 

of this section, which is used to send WordBasic statements from SAS to Word to implement actions in 

Word. After starting Word, the program pauses for 3 seconds ❶ for Word to open before setting up 

the connection in FILENAME statement. Otherwise, the following DATA steps may have running 

errors. WordBasic statement FileSaveAs with option 6 ❷ specifies RTF as the file format to be saved. 

Then, function Call Execute ( ) ❸ runs macro %word2rtf through all the Word files in the dataset 

fileList. After this step, the RTF folder will be filled with converted RTF files. 

3. Subtract data from RTF files 
 

******************************* 

* Read in data from RTF files * 

*******************************; 

%macro rtf2sas(rtfFile,fileName,surveyNum); 

fileName inRTF "&rtfFile"; 

 

data questions; 

   infile inRTF lrecl=5000 truncover; 

   input rawTxt $ 1-500; 

 

   retain Criterion Keep; 

 

   if index(rawTxt, " 1: ") then do;                        ❶a 
 Criterion=1; 

 Keep=1; 

   end; 

 

   if index(rawTxt, " 2: ") then Criterion=2; 

   if index(rawTxt, " 3: ") then Criterion=3; 

   if index(rawTxt, " 4: ") then Criterion=4; 

   if index(rawTxt, " 5: ") then Criterion=5; 

   if index(rawTxt, " 6: ") then Criterion=6; 

   if index(rawTxt, " 7: ") then Criterion=7; 

   if index(rawTxt, " 8: ") then Criterion=8; 

 

   if Keep=1; 

   if index(RawTxt, " 1: ") & lag(Criterion)=8 then stop;   ❶b 
   drop Keep; 

run; 

 

data q_and_a; 

   set questions; 

   by Criterion; 

 

   if first.Criterion then checkBox=0; 

   if index(rawTxt,"0000140000000") then do;                              ❷ 

 checkBox+1; 

 if index(rawTxt,"00010000000000000000000000") then checkBoxValue=1; ❸ 

    else checkBoxValue=0; 

 output; 

 end; 

run;  

 

data selected; 

   set q_and_a; 

   if checkBoxValue=1; 

run; 

proc transpose data=selected out=record(drop=_name_) prefix=Criterion; 

   id Criterion; 

   var checkBox; 

run; 

 

data survey_&surveyNum; 

   length Survey $ 50; 

   set record; 

   Survey="&fileName"; 

run; 

%mend rtf2sas; 

 

data _null_; 

   set fileList; 

   call execute('%rtf2sas('||rtfPathname||','||fileName||','|| 

                 fileID||')');                                     ❹ 

run; 

 

 

**************************** 

* Set all surveys together * 

****************************; 

data output.Survey_All;                                            ❺ 

 set survey_:; 

run; 

 

From ❶a to ❶b, the code is to assign values to variable Criterion according to the unique criterion 

identification numbers. And in the meantime, the rawTxt values before and after question-and-answer 

segment is dropped. The variable Keep is used to get rid of all RTF contents before the first question; 

and statement of ❶b is used to get rid of all RTF contents after the last question by using next “1: “ 

(start of next section that is not a part of data needed).  For very checkbox, there is a value portion 

“{\*\datafield 650000001400000007436865636b353500010000000000000000000000}”. In this value, 

“0000140000000” ❷ is used to identify any checkbox; “00010000000000000000000000” ❸ indicates 

the box is checked, and “00000000000000000000000000” indicates the box is blank. Then, the 

observations with “checkBoxValue = 1” are kept, and transposed with the “Criterion underscore 

criterion number” as variable names to change the data layout from vertical to horizontal style. To 

execute the macro, function Call Execute ( ) ❹ runs %rtf2sas through all questionnaire files in RTF 

folder. At last, all individual datasets are set together ❺ to generate the final output SAS dataset 

shown in Figure 1. 

 

COMMENTS 
 
The way to collect study data described above is a little unconventional. However, if you have no 

control on the data collection, and what you are requested to do is to get the data from Word files into 

a SAS dataset, using SAS program to read in data directly from the raw Word files can save a lot of 

intermediary efforts, such as building up a database, entering data, and then importing data into SAS 

system. In terms of efficiency, this SAS programming approach is a shortcut. It can also avoid some 

human errors during data entry. 

The programming in this paper is not complicated. The key part is to find out the unique structure 

characters in RTF file to identify the data to be subtracted. In other similar situations, the specifics can 

be quite different. The programming always needs to be tailed to the situation at hand, however, the 

basic idea for importing data without table structure should be similar: first, convert raw files to SAS 

readable if needed; then read it into SAS, and subtract data by the file structure specifications.  

